Dating Isn't Really That Hard!

HelpMeExcel.com

James Tobin Consulting, LLC

Some Stuff Before We Get Started…

Quick Tips!!!	Keyboard Combo	Results	Comments
Quick Date Inserts	Ctrl + :	9/12/2018	Non-Volatile
Quick Time Inserts	Ctrl + Shift + :	10:55 AM	Non-Volatile

****** For demonstration purposes, the following "Named" dates are used: ******

Named Range	Date	Is the Date Text?
TheDate ==>	9/30/2018	FALSE
TheDateText ===>	12/31/2001	TRUE

Date "Parts"

If you want to display the:	Formula	Current Application	Result	Comments
Month	=MONTH(any numeric date)	=MONTH(TheDate)	9	
Day	=DAY(any numeric date)	=DAY(TheDate)	30	
Year	=YEAR(any numeric date)	=YEAR(TheDate)	2018	
Date	=DATE(year,month,day)	=DATE(TheYear,TheMonth,TheDay)	9/30/2018	Reverse Engineered Date
Weekday	=WEEKDAY(any numeric date)	=WEEKDAY(TheDate)	1	In Excel, Weekday 1 = Sunday
Now	=NOW()	=NOW()	1/9/2019 13:27	Volatile; System Driven; Might Slow Workbook Speed
Today	=TODAY()	=TODAY()	1/9/2019	Volatile; System Driven; Might Slow Workbook Speed
End of the Month	=EOMONTH(numeric date,month-end offset)	=EOMONTH(TheDate,-4)	5/31/2018	0 = The Current Month's End
Date's value	=DATEVALUE(any text date)	=DATEVALUE(TheDateText)	37256	
Difference btwn Dates	=DAYS(end date, start date)	=DAYS(TheDate,TheDateText)	6,117	Compare to DateDif "d"
Difference btwn Dates	=DATEDIF(lesser numeric date, greater numeric date,"d" or "m" or "y") yes, in quotes…	=DATEDIF(DATEVALUE(TheDateText),TheDate,"m")	200	"m" = Months; full months only
				**** Proof ****
			192	16 full years times 12 months
			8	Jan - Sep 2018
		Proof	200	
		=DATEDIF(DATEVALUE(TheDateText),TheDate,"d")	6,117	"d" = Days
				**** Proof ****
			5,840	16 full years times 365 days
			4	2004 2008 2012 2016 Leap Years
			273	Number of days from Jan 1 thru Sep 30, 2018
		Proof	6,117	
		=DATEDIF(DATEVALUE(TheDateText),TheDate,"y")	16	"y" = Year; complete years only
				**** Proof ****
		Proof	16	Full years from 2001 thru 2018

Dating Isn't Really That Hard!

HelpMeExcel.com

James Tobin Consulting, LLC

Text Formatting

You can present a day/date in a text string:

Basic Formula: ===> =Text(any numeric date, "format text") yes, in quotes…

	Text Formatting Options	Current Application	Result
For Dates: ===>	"m" = 1 or 2 numeric characters for the Month.	=TEXT(TheDate,"m")	9
	"mm" = 2 numeric characters for the Month. January for example, would be 01.	=TEXT(TheDate,"mm")	09
	"mmm" = First 3 Alpha characters of the Month.	=TEXT(TheDate,"mmm")	Sep
	"mmmm" = Full Alpha name of Month.	=TEXT(TheDate,"mmmm")	September
	"d" = 1 or 2 digits for the Day.	=TEXT(TheDate,"d")	30
	"dd" = 2 digits for the Day. The fourth day of the month for example, would be 04.	=TEXT(TheDate,"dd")	30
	"ddd" = First 3 characters of the Alpha Day of the Week. This must preceed the mm/dd/yyyy treatment.	=TEXT(TheDate,"ddd")	Sun
	"dddd" = Full Alpha Day of the Week. This must preceed the mm/dd/yyyy treatment.	=TEXT(TheDate,"dddd")	Sunday
	"yy" = Last 2 digits of the Year.	=TEXT(TheDate,"y")	18
	"yyyy" = All four digits of the Year.	=TEXT(TheDate,"yyyy")	2018
	Fully Blown Out!!	=TEXT(TheDate,"dddd mmmm dd, yyyy")	Sunday September 30, 2018
	Nested formulas with mixed formats!!	="For the "&MONTH(TheDate)&" Months Ending "&TEXT(TheDate,"mmmm dd, yyyy")&"." The "&" tells MSExcel to toggle from "Text" to "Formula" or visa-versa. *****!!! OR !!!***** =CONCATENATE("For the ",MONTH(TheDate)," Months Ending ",TEXT(TheDate,"mmmm dd, yyyy"),".")	For the 9 Months Ending September 30, 2018. For the 9 Months Ending September 30, 2018.

Custom Formatting

Custom Formats can simplify and stabilize presentations.

A possible drawback to "text" formatting is that it might require a "helper" cell or reference. In other words, the "=Text()" formula might require a reference to another cell or data point that contains the date.

To resolve the drawback, you can use a "Custom" format. If the cell contains a custom format, a "helper" cell is not required. Simply input the date in the presentation cell and the custom formatting will present the date according the custom format you've created.

For example, if you wanted to ensure that the date presentation for your comapany's quarterly Income Statement is always correct, you'll need to create a custom format via the Format Cells dialogue box.

To create the custom formatting, right mouse click and open the Formatting Cells Dialogue box.

In the "Category:" section, scroll to the bottom and select "Custom".



In the "Type:" field, type the custom formatting you wish to use. If the custom formatting begins with text, the custom format must begin with double quotes. Also, each introduction of a number must begin and end with double quotes.

In this particular case, you'd place the following into the "Type:" field (yes include the quotes): "For the "m" Months Ending "mmmm dd, yyyy"."

Hit the "OK" button to complete the creation of the custom format.

As an example, input "9/30/2018" (sans quotes) into the cell where you'd placed the custom format (e.g. the date presented on the Income Stmt).

Apply the custom formatting and….

James Tobin Consulting, LLC
Income Statement
For the 9 Months Ending September 30, 2018.